
<job> 
<script language="VBScript"> 
 
'--------------------------------------------------------- 
' This sample launches the application as interactive user. 
'--------------------------------------------------------- 
 
const TriggerTypeRegistration = 7 ' A constant that specifies a registration trigger. 
const ActionTypeExecutable = 0 ' A constant that specifies an executable action. 
const FlagTaskCreate = 2  ' A constant that specifies the flag in RegisterTaskDefinition. 
const LogonTypeInteractive = 3 ' A constant that specifies an executable action. 
 
If WScript.Arguments.Length <> 1 Then 
WScript.Echo "Usage: cscript launchapp.wsf <AppPath>"  
WScript.Quit 
End If 
 
strAppPath = WScript.Arguments(0) ' This is the script to run. 
 
'******************************************************** 
' Is this Vista or XP? 
' Begin the section added by Jarvis 
'******************************************************** 
' This section that determines OS version is a modified version of code 
' originally posted at the following web page: 
' http://blog.eqinox.net/jed/archive/2006/12/05/1270.aspx 
' 
Dim vistaClient 
Dim wshShell 
Set wshShell = CreateObject("WScript.Shell") 
 
computer = "." 
Set objWMIService = GetObject("winmgmts:{impersonationLevel=impersonate}!\\" & computer & "\root\cimv2") 
Set objOSList = objWMIService.ExecQuery("Select * from Win32_OperatingSystem") 
For Each os In objOSList 
 BuildNumber = OS.BuildNumber  
 If BuildNumber < 6000 Then  
  NotVistaClient = True 
 End If 
Next 
 
If NotVistaClient = True Then 
   wshShell.Run(strAppPath), 0, False ' Run the logon script normally 
   WScript.Quit 
Else    ' Run the logon script using launchapp.wsf 
 
'******************************************************** 
' The next two lines set a string variable that adds two minutes to the 
' current date/time to be used to set an endboundary to the scheduled task. 
' This uses a Function named format that is at the end of the script 
'******************************************************** 
 
strTwoMinutesfromNow = format(now(), "yyyy-mm-dd")  
strTwoMinutesfromNow = strTwoMinutesfromNow & "T" & format(DateAdd("n", 2, time()), "hh:mm:ss") 
 
'******************************************************** 
' This function was attributed to Scott Dixon on the following web page 
' http://classicasp.aspfaq.com/date-time-routines-manipulation/can-i-make-vbscript-format-dates-for-me.html 
' 
Function Format(vExpression, sFormat)  
        set fmt = CreateObject("MSSTDFMT.StdDataFormat")  
        fmt.Format = sFormat  
        set rs = CreateObject("ADODB.Recordset")  
        rs.Fields.Append "fldExpression", 12 ' adVariant  
        rs.Open  



        rs.AddNew  
        set rs("fldExpression").DataFormat = fmt  
        rs("fldExpression").Value = vExpression  
        Format = rs("fldExpression").Value  
        rs.close: Set rs = Nothing: Set fmt = Nothing  
End Function  
'******************************************************** 
 
'******************************************************** 
' End the section added by Jarvis 
'******************************************************** 
 
'******************************************************** 
' Create the TaskService object. 
'******************************************************** 
 
Set service = CreateObject("Schedule.Service") 
call service.Connect() 
strTaskName = "Launch Logon Script As Interactive User" 
 
'******************************************************** 
' Get a folder to create a task definition in.  
'******************************************************** 
 
Dim rootFolder 
Set rootFolder = service.GetFolder("\") 
'Delete the task if already present 
On Error Resume Next 
call rootFolder.DeleteTask(strTaskName, 0) 
Err.Clear 
 
'******************************************************** 
' Create the new task 
'******************************************************** 
 
Dim taskDefinition 
Set taskDefinition = service.NewTask(0)  
 
'******************************************************** 
' Create a registration trigger. 
'******************************************************** 
 
Dim triggers 
Set triggers = taskDefinition.Triggers 
Dim trigger 
Set trigger = triggers.Create(TriggerTypeRegistration)  
trigger.EndBoundary = strTwoMinutesfromNow 
 
'*********************************************************** 
' Create the action for the task to execute. 
'*********************************************************** 
 
' Add an action to the task. The action executes the app. 
 
Dim Action 
Set Action = taskDefinition.Actions.Create( ActionTypeExecutable ) 
Action.Path = strAppPath 
WScript.Echo "Task definition created. About to submit the task..." 
 
'*********************************************************** 
' Register (create) the task. 
'*********************************************************** 
 
call rootFolder.RegisterTaskDefinition(strTaskName, taskDefinition, FlagTaskCreate,,, LogonTypeInteractive) 
WScript.Echo "Task submitted." 



 
End If 
 
</script> 
</job> 


